
PARTE I

IL SISTEMA SAS

Sistema integrato di prodotti software

• data entry, manipolazione archivi;

• stesura di report e grafici;

• analisi statistiche e matematiche;

• previsioni e supporto alle decisioni;

• ricerca operativa e project management;

• sviluppo di applicazioni.

Il fulcro del sistema SAS è il modulo SAS Base

• linguaggio SAS;

• procedure per analisi dei dati e stesura di report;

• macro-linguaggio.

ORGANIZZAZIONE DEL SISTEMA SAS

L’analisi dei dati si svolge seguendo due passi fondamentali:

1)  L’organizzazione dei dati

2)  L’analisi dei dati 

Nel SAS System

Data Step

Inizia con un data statement e consente di leggere/creare e/o modificare archivi di dati

Proc Step

Inizia con un proc statement, ovvero richiama una procedura SAS, la esegue su un SAS dataset  e produce dei risultati

NB la suddetta suddivisione è una semplificazione perché in realtà vedremo che anche il proc statement può creare un data set

Un programma SAS può essere costituito da 

· Un solo Data Step

· Un solo Proc Step

· Più  Data Step e/o più Proc Step

AMBIENTE DI LAVORO A FINESTRE

Le finestre base del SAS per Windows sono 5:

· SAS  Explorer 
contenuto delle LIBRARIES e dei SAS datasets;

· SAS Results visualizzazione dei risultati di ogni procedura eseguita;

e le finestre di programmazione

· SAS Enhanced Editor e Program Editor scrittura istruzioni (programmi);

· SAS  Log  esito operazioni eseguite. Una volta eseguito il programma, SAS scrive dei messaggi nel SAS LOG, ignorare tali messaggi può essere pericoloso perché talvolta si ottengono dei risultati ma tali risultati potrebbero essere non corretti per qualche problema intercorso nelle istruzioni digitate;

· SAS  Output risultati delle esecuzioni.

Inoltre:

SAS SYSTEM HELP per la consultazione della sintassi e delle opzioni del linguaggio e delle procedure

Visualizzazione dei risultati nella finestra di OUTPUT

Una volta ‘sottomesso’ il programma SAS (SAS windowing environment), i risultati appariranno nella finestra OUTPUT  e nella finestra Results window.

Printing or saving the contents of the Output window 

If you want to print or save the entire contents of the Output window, first make the Output window active by clicking in it, then select either Print or Save As from the File pull-down menu. If you cannot print from within SAS, then save the output to a file and use your system’s command for printing files  

The Results window

When you have a lot of output, the Results window can be very helpful. The Results window is like a table of contents for your output. It lists each procedure that produces output, and if you open, or expand, the procedure in the Results tree, you can see each part of the procedure output. The following figure shows what your screen might look like if you ran the ANOVA (Analysis of Variance) procedure.

There is one entry in the Results window for the ANOVA procedure. If you expand the ANOVA procedure in the results tree, by clicking on the plus (+) signs, then you will

see all the different parts of the ANOVA output. Double click on the output you want to see, and it will appear at the top of the Output window. The following figure shows what your Output window would look like after you double click on the Overall ANOVA item in the Results window.

Printing or saving parts of the output

Using the Results window, it is possible to print or save just the parts of the output you want. First highlight the item you want in the Results window, then bring up the context-sensitive menu.

In the Windows operating environment you do this with

the right mouse button. Then select either Print or Save As from the pop-up menu. 

Creating HTML Output

If you are using the SAS windowing environment, then you can create output in Hypertext Markup Language (HTML) format with just a few clicks of your mouse.

Select Options-Preferences from the Tools menu. This opens the Preferences window. Click on the Results tab to bring it to the front. Here is what the Results portion of the Preferences window looks like in Windows:

[image: image4.wmf]
When you first open this window, you will see a check next to Create Listing. Listing is the default type of output, and it is what you see in the Output window if you are using the SAS windowing environment. You can turn on HTML output by clicking in the box next to Create HTML. To turn off the listing or HTML output, just click to un-check it. Once you have turned on HTML output, then every time you run a program, your output will automatically appear in the Results Viewer window.

SAS DATA SET

Data Step e Proc Step si applicano ai file in formato SAS - SAS DATA SET -

I  SAS DATA SET sono organizzati in forma rettangolare 

· ogni riga rappresenta un’osservazione; 

· ogni colonna rappresenta una variabile
· tutte le osservazioni possiedono le stesse variabili( le variabili non osservate per una data osservazione sono registrate come mancanti (missing)

Ogni SAS dataset è autodescrittivo, infatti esplicita: 

· Il nome del SAS data set
· Il nome delle variabili, le label, ed il formato

· Il contenuto delle variabili per ogni osservazione

Dimensione del  SAS data set 

Prior to SAS 9.1, SAS data sets could contain up to 32,767 variables.

Beginning with SAS 9.1, the maximum number of variables in a SAS data set is limited by the resources available on your computer. 

The number of observations, no matter which version of SAS you are using, is limited only by your computer’s capacity to handle and store them.

LIBRERIE: DEFINIZIONE E GESTIONE

Before you can use a SAS data set, you have to tell SAS where to find it. You do that by setting up a SAS library. A SAS library is simply a location where SAS data sets are stored. A SAS library might be a folder or directory on your computer, or it might be a physical location like a hard drive, or CD. 

Per definire una libreria permanente si possono seguire due strade:

· fare click sul comando New Library della barra Menu ed inserire le informazioni richieste, tra cui il nome della library ed il percorso fisico associato alla libreria stessa

· Scrivere nella finestra Program Editor l‘istruzione

LIBNAME  mylibname  ‘path’;

Dove:

mylibname 
è il nome attribuito alla libreria

path 
è il percorso fisico dove vengono memorizzati i dati

The Active Libraries window

· When you open the Active Libraries window, you will see at least three libraries: SASHELP; SASUSER; WORK. You may have other libraries for specific SAS products (such as the Maps library for SAS/GRAPH software), or libraries that have been set up by you or someone you work with. 

· The Sashelp library contains information that controls your SAS session along with sample SAS data sets.

· The Work library is a temporary storage location for SAS data sets. It is also the default library. If you create a SAS data set without specifying a library, SAS will put it in the Work library, and then delete it when you end your session. 

· If you make changes to the default settings for the SAS windowing environment, this information will be stored in the Sasuser library.

IL LINGUAGGIO SAS: concetti introduttivi

In SAS vengono utilizzati SAS statement  per scrivere una serie di istruzioni (statement) che vanno a costituire il SAS PROGRAM.

Naturalmente per scrivere un programma SAS è necessario utilizzare il linguaggio appropriato il SAS LANGUAGE.

Il SAS PROGRAM è costituito da una serie di istruzioni ordinate.

La prima regola da seguire nella stesura di un programma è:

every SAS statement ends with a semicolon
Istruzioni SAS (SAS statement)
· Iniziano con una parola chiave

· Terminano sempre con il carattere punto e virgola

· Possono essere scritte con lettere maiuscole e minuscole, il linguaggio SAS non è case sensitive

· Possono iniziare in una qualsiasi colonna della riga e proseguire su più righe

· Più istruzioni possono essere scritte sulla stessa riga ma devono essere separate da punto e virgola

Comments

There are two styles of comments you can use: 

· one starts with an asterisk (*) and ends with a semicolon (;) the other style starts with a slash asterisk (/*) and ends with an asterisk slash (*/).

Per salvare un programma SAS si fa uso del Menu principale. Il programma viene salvato con estensione .SAS
REGOLE PER NOMI SAS DEFINITI DALL’UTENTE

· Numero massimo di caratteri dipende dal tipo di nomi (32, 8)

· Il primo carattere deve essere una lettera o un underscore “_”

· I caratteri successivi al primo possono essere lettere, cifre, “_”

· Indipendentemente da maiuscolo o minuscolo, SAS converte tutto in maiuscolo

· Nei nomi non  possono comparire spazi bianchi

· Non sono ammessi caratteri speciali ($,£,*…)

· Non si possono utilizzare nomi di variabili automatiche del sistema (_N_, ERROR_)

· Non si possono utilizzare nomi che il SAS riserva a librerie speciali (LIBRARY, SASHELP, WORK, USER, MAPS..)

ESECUZIONE PROGRAMMI SAS

Esecuzione in modo semi-interattivo 

· Stesura del programma nella finestra Enhanced/Program editor

· Visualizzazione risultati nella finestra Output

· Segnalazioni inviate dal sistema (messaggi di errore, warning, altro)  nella finestra Log

Il programma SAS viene automaticamente compilato dal sistema prima di essere eseguito:

· compilazione ed esecuzione avvengono a blocchi  (data step, proc step);

· i blocchi si chiudono con la parola chiave RUN; 

INTRODUZIONE AL DATA STEP

La struttura del Data Step è la seguente:

DATA nomefile (opzioni);
…..

RUN;

Il DATA step esegue il programma linea per linea e osservazione per osservazione

Significa che  

· il SAS legge l’obs numero 1 del data set

· esegue tutte le istruzioni del Data step 

· se non incontra errori scrive l’obs  numero 1 del data set di output

· torna all’inizio del data step per operare sull’obs numero 2 e prosegue come in un loop fino all’ultima obs.

L’ISTRUZIONE DATA e L’ISTRUZIONE SET

DATA nomedataset_new <opzioni>;

SET nomedataset_old;

RUN;
Dove nomedataset_new/old è il nome di un Sas dataset. Tale nome può essere scritto a due livelli:

nome1.nome2

· nome1: è il nome di primo livello ed indica la libreria in cui memorizzare il Sas data set (per default la libreria è WORK)

· nome2: è il nome del Sas data set che viene memorizzato nel percorso fisico associato alla libreria.

Come opera il suddetto blocco di istruzioni?

· legge il file indicato all’istruzione SET

· scrive sul file indicato all’istruzione DATA
PARTE II

LETTURA DI DATI DI TIPO ASCII

· Istruzione INFILE

· Istruzione INPUT

· Istruzione DATALINES (CARDS)

Istruzione INFILE

Indica al sistema dove leggere i dati 

INFILE ‘nomefile’ [opzioni]; 
nomefile 

nome, con eventuale percorso, del file ASCII da leggere o parola chiave CARDS se i dati sono inseriti da programma

Istruzione INPUT

Definisce nome, tipo e modo di lettura delle variabili 

I modi di lettura sono: a lista, a colonna, con formato

LETTURA A LISTA
Possibile quando i dati sono registrati in formato libero, con almeno uno spazio bianco tra un campo ed il successivo

INPUT var1 var2 var3;
INPUT var1-var10;
INPUT var1 $ var2;

Lettura a lista da file esterno
Input dataset: INPUT_LISTA.TXT

	100 1 34

	200 1 65

	300 2 29

	400 1 31

	500 1 45

	600 2 40

	700 2 68

	800 1 51

	900 1 48


data pippo;

infile 'F:\written\didattica\CorsoSAS\input_lista.txt';

input codice genere  eta;

run;
Lettura a lista e scrittura del file da programma

data pluto; 











input x1-x3;

datalines;

1 5 7

9 3

2

6 9 8 

13 5 8

;

run;

equivalentemente

input x1-x3;

cards;

1 5 7

9 3

2

6 9 8 

13 5 8

;

run;

Output SAS data set PLUTO

	x1
	X2
	X3

	1
	5
	7

	9
	3
	2

	6
	9
	8

	13
	5
	8


LETTURA A COLONNA
Input dataset: INPUT_COLONNA.TXT

	100m34

	200m65

	300f29

	400m31

	500m45

	600f40

	700f68

	800m51

	900m48

	999m36


data pippo;

infile 'F:\written\didattica\CorsoSAS\input_colonna.txt';

input codice 1-3 genere $ 4 eta 5-6;

run;

LETTURA CON FORMATO

Non si specificano le colonne ma la lunghezza di ogni campo

INPUT var informat.  ;

INPUT (varlist) ( informat list ) ;

INPUT (varlist) ([n*] informat.) ;

Sintassi generale informat

w. 
legge numeri interi o decimali con punto decimale codificato nel campo

$w. 
Legge stringhe di caratteri ASCII

*LETTURA A FORMATO: i campi non sono separati;
data pippo;

infile 'F:\written\didattica\CorsoSAS\input_colonna.txt';

input codice 3.  genere $1.  eta 2.;

run;

In questo caso i dati sono separati da uno spazio quindi devo incrementare il puntatore 

*LETTURA A FORMATO: i campi sono separati da uno spazio;
data pippo;

infile 'F:\written\didattica\CorsoSAS\input_lista.txt';

input codice 3. +1 genere $1. +1 eta 2.;

run;

alternativamente il simbolo @i indica al puntatore di spostarsi alla colonna i

data pippo;

infile 'F:\written\didattica\CorsoSAS\input_lista.txt';

input codice 3. @5  genere $1. @7 eta 2.;

run;

Si tenga presente che la lettura con formato è indispensabile nel caso in cui ci siano da leggere dei campi contenenti una data.

Esempio: Supponiamo di avere dei dati provenienti dalla degustazione di 4 vini di marche diverse. I degustatori sono 3. I dati inseriti nel file sono: la marca del vino, l’anno di produzione, la data di degustazione, i punteggi dei tre degustatori.

*Esempio di lettura con formato ;
data punti_vino;

infile cards;

input marca $1 +1 anno 2. +1 data ddmmyy10. +1 (punti1-punti3) (3*4.);

cards;

A 93 20-11-1994 7.8 7.1 6.5

B 95 23-12-1998 7.9 7.6 7.5

C 99 10-11-2000 6.5 7.0 6.8

D 00 30-10-2002 5.9 6.4 7.2

;

run;

proc print data=punti_vino;

var marca anno punti1-punti3 data;

format data date8.;

run;

proc print data=punti_vino;                                                                                                             

var marca anno punti1-punti3 data;                                                                                                      

format data MONyy.;                                                                                                                     

run;                                                                                                                                    

proc print data=punti_vino;                                                                                                             

var marca anno punti1-punti3 data;                                                                                                      

format data DDMMyy.;                                                                                                                    

run;                      

LETTURA DI PIU’ OSSERVAZIONI DA UNO STESSO RECORD

@@    indica la reale fine del record, quindi nell’esempio che segue i dati vengono letti a coppie

/*LETTURA DI PIU' OSSERVAZIONI DA UNO STESSO RECORD*/
data unrecord;

input genere $ peso @@;

cards;

m 60 f 50 m 68

m 82 f 55 f 56

;

run;

LETTURA DI UNA OSSERVAZIONE SU PIU’  RECORDS

/
indica di andare al record successivo

#n  
indica di andare al record n

Nell’esempio l’osservazione relativa a ciascun vino è registrata su 3 record:

· nel 1°  la marca del vino, l’anno di produzione, 

· nel 2°  la zona di provenienza del vino,

· nel 3°   il punteggio attribuito al vino.

/*LETTURA DI una  OSSERVAZIONE SU PIU' RECORD*/
*ESEMPIO VINO, ELIMINATA LA DATA;
data punti_vino_zona;

infile cards;

input marca $ anno  / ZONA $  #3 punti;

cards;

A 1993 

ZONA1

7 

B 1995

ZONA2

7.9 

C 1999

ZONA3

6.5 

D 2000

ZONA4

5.9 

;

RUN;

DATI NEI RECORD E VARIABILI IN INPUT

(a) le variabili specificate esauriscono completamente i dati presenti nel record;

(b) le variabili specificate richiedono un numero di dati minore rispetto a quelli esistenti;

(c) le variabili specificate richiedono un numero di dati superiore rispetto a quello esistente.

(c) SAS prosegue nell’input del primo record leggendo dalla riga seguente; su tale riga salta i successivi campi e va a capo a leggere nel record successivo partendo dalla prima colonna e scrive sulla finestra LOG il messaggio:

NOTE: SAS went to a new line when INPUT statement reached past the end of a line.

data uno;

input var1-var6;

cards;

18 66 24 23

19 45 34 20

40 97 56 43

23 56 73 90

;

run;

NOTA: SAS è passato a una nuova riga quando l'istruzione INPUT ha superato la fine di una riga.

NOTA: Il data set WORK.UNO ha 2 osservazioni e 6 variabili.

Oss  var1    var2    var3    var4    var5    var6

1      18      66      24      23      19      45

2      40      97      56      43      23      56
CONTROLLO ESAURIMENTO DATI

INFILE nome_file MISSOVER;

Fa sì che la lettura non vada oltre la fine del record specificato nell’istruzione cards, assegna valore mancante a quelle variabili per cui non vi sono dati.

Data tre;

infile cards missover;

cards;

18 66 24 23

19 71 26 21

20 68 23 29

;

run;

proc print data=tre;

run;
NOTE: The data set WORK.TRE has 3 observations and 6 variables.

NOTE: The DATA statement used 0.0 seconds.

OBS 
VAR1 
VAR2 
VAR3 
VAR4 
VAR5 
VAR6

1 

18 

66 

24 

23 

.

.

2 

19

71

26 

21 

. 

.

3 

20

68

23

29 

.

.
INFILE nome_file TRUNCOVER;

Fa sì che la lettura non vada oltre la fine del record assegna valore mancante a quelle variabili per cui non vi sono dati.

TRUNCOVER consente di leggere records a lunghezza variabile quando alcuni records sono più corti di quanto

dichiarato nell’istruzione INPUT.

Data tre;

input var1-var6;

infile cards;

cards;

18 66 24 23 12 1

19 71 26 21 3

20 68 23 29

1

;

run;

proc print data=tre;

run;

NOTA: LOST CARD.

RIGHELLO:  ----+----1----+----2----+----3----+----4----+----5----+----6----+----7----+----8--

44         ;

var1=1 var2=. var3=. var4=. var5=. var6=. _ERROR_=1 _N_=3

NOTA: SAS è passato a una nuova riga quando l'istruzione INPUT ha superato la fine di una riga.

NOTA: Il data set WORK.TRE ha 2 osservazioni e 6 variabili.
Data tre;

input var1-var6;

infile cards truncover;

cards;

18 66 24 23 12 1

19 71 26 21 3

20 68 23 29

1

;

run;

proc print data=tre;

run;

NOTA: Il data set WORK.TRE ha 4 osservazioni e 6 variabili.
Oss    var1    var2    var3    var4    var5    var6

1      18      66      24      23      12       1

2      19      71      26      21       3       .

3      20      68      23      29       .       .

4       1       .       .       .       .       .
SCRITTURA DEL DATASET SULLA FINESTRA DI OUTPUT

/*Istruzione PUT per scrivere sulla finestra di output il data set*/
data report;

input id 1-2 num 3-6;

file print;

if _N_=1 then put @20 'Lista del Comune di Siena';

put @5 'obs #' id '=' num;

if _N_=5 then put   @20 'Lista del Comune di Firenze';

cards;

10123

11345

12456

13678

14789

21543

22478

;

run;

IMPORTAZIONE/ESPORTAZIONE AUTOMATICA DI DATI 

Dal Menu principale:

· File 

· Import Data/Export Data

· ….

* tracciato record indagine consumi ISTAT

*esempio di importazione da EXCEL dati  Istat “indagine sui Consumi delle famiglie italiane”, il file CONSUMI_TOSCANA.XLS è una selezione di variabili per la sola Regione Toscana

Se l’operazione di importazione ha funzionato nella finestra di Log apparirà:

NOTE: WORK.CONSUMO_TOSCANA was successfully created

IL LINGUAGGIO SAS: PROCEDURE GENERALI RELATIVE ALLA VISUALIZZAZIONE  DEI  DATA SET

· PROC CONTENTS 

· PROC PRINT 

· PROC FORMAT

*PROGRAM_PROCEDURE_GENERALI.SAS;
PROC CONTENTS: permette di visualizzare molte informazioni sul Sas data set, tra cui la directory della libreria SAS e l’elenco di tutte le variabili in ordine alfabetico.

PROC CONTENTS  <options>;

Con l’opzione VARNUM  la lista delle variabili rispetta l’ordinamento fisico del Sas data set

PROC PRINT: stampa le osservazioni del Sas data set relative a tutte o ad una selezione di variabili. Si possono creare report semplicissimi (elenco osservazioni) o anche complessi utilizzando le varie opzioni.

PROC PRINT < option(s)>; 

	
	BY < DESCENDING> variable-1 

<...< DESCENDING> variable-n>< NOTSORTED>; 

PAGEBY BY-variable; 

SUMBY BY-variable; 




	
	ID variable(s); 


	
	SUM variable(s); 


	
	VAR variable(s); 


/*inizia dalla osservazione 5 e stampa 10 osservazioni*/
proc print data= consumo_toscana (firstobs=5 obs=10);

var numcomp sesso1 eta1;

run;

/*inizia dalla osservazione 1 e stampa 10 osservazioni, inserendo la SUM*/
proc print data= consumo_toscana (obs=10);

sum bar;

var numcomp sesso1 eta1 bar;

run;

PROC FORMAT: consente di creare una libreria di format permanenti da utilizzare nei data step o nelle procedure

PROC FORMAT <option(s)>; 

…..

	
	VALUE <$>name <(format-option(s))> 
value-range-set(s); 


******************PROC FORMAT*******************;
proc format;

value $genere 'm'='maschio'
              'f'='femmina';

value si_no    1='sì'
               2='no';

run;

data segreteria;

input      id 1. @3 genere $1. @5 ricovero 1.  @7 analisi  1.;


  format genere   $genere.
             ricovero si_no.



  analisi  si_no.;

cards;

1 m 1 1

2 f 2 1

3 f 1 2

4 m 2 1

  ;

  run;

  proc print;run;

How do I read a SAS data file when I don't have its format library? 

If you try to use a SAS data file that has permanent formats but you don't have the format library, you will get errors like this.

ERROR: The format $MAKEF was not found or could not be loaded.

ERROR: The format FORGNF was not found or could not be loaded.

Without the format library, SAS will not permit you to do anything with the data file. However, if you use 

OPTIONS nofmterr;

at the top of your program, SAS will go ahead and process the file despite the fact that it does not have the format library. You will not be able to see the formatted values for your variables, but you will be able to process your data file 

LINGUAGGIO SAS: GESTIONE  DEI  DATA SET

· LABEL

· RENAME

· KEEP

· DROP 

· RETAIN 

*PROGRAM_GESTIONE1;
LABEL: consente di assegnare nomi estesi alle variabili la label non può superare 40 caratteri;

LABEL variable = "label" ... ;

RENAME: consente di modificare il nome della variabile;

RENAME vecchio_nome = nuovo_nome. ;

DROP: elenca le variabili da eliminare nel SAS data set;

DROP  variabili ;

KEEP: elenca le variabili da scrivere nel SAS data set;

KEEP  variabili ;

Esempio: seleziono solo le variabili che riguardano il capofamiglia

data lib.consumo_toscana;set consumo_toscana;

run;

data info_capof (keep=rela1 genere1 eta1 statociv1 titstu1 conprof1 posprof1);

set lib.consumo_toscana;

label sesso1=sesso capofamiglia eta1=eta capofamiglia;

rename sesso1=genere1 ;

run;

N.B. Nel data set info_capof le variabili non sono nell’ordine specificato, se voglio che il data set rispetto l’ordinamento delle variabili che io specifico devo usare l’istruzione retain

data info_capof (keep=rela1 genere1 eta1 statociv1 titstu1 conprof1 posprof1);

retain  rela1 genere1 eta1 statociv1 titstu1 conprof1 posprof1;
set lib.consumo_toscana;

label sesso1=sesso capofamiglia eta1=eta capofamiglia;

rename sesso1=genere1 ;

run;

PARTE III

LINGUAGGIO SAS: ISTRUZIONI DI ASSEGNAZIONE

Variabile=Espressione

L’espressione è una sequenza di: 

· operandi (variabili, costanti);
· operatori (caratteri speciali, funzioni, parentesi)

Tipo di operatori:

Aritmetici, di comparazione, logici, carattere

Regole ordine di esecuzione

I. Espressioni entro parentesi

II. 7 livelli di priorità (1=massima)

OPERATORI ARITMETICI

	priorità
	simbolo
	descrizione

	1
	**
	Elevamento a potenza

	2
	*
	moltiplicazione

	2
	/
	Divisione

	3
	+
	addizione

	3
	-
	sottrazione


Gli operatori aritmetici

· Agiscono su variabili di tipo numerico;

· Conversione automatica da carattere a numerico;

· Gli operatori con uguale priorità vengono eseguite da sin a dx; l’elevamento a potenza da dx a sin;

· le parentesi possono modificare la priorità degli operatori;

· tutte le operazioni aritmetiche vengono eseguite in doppia precisione;

OPERATORI DI COMPARAZIONE

	<
	LT
	Minore di

	<=
	LE
	Minore o uguale di

	>
	GT
	Maggiore di

	>=
	GE
	Maggiore o uguale di

	^=
	NE
	Non uguale

	
	IN
	Uguale a uno degli elementi della lista


Gli operatori di comparazione:

· effettuano un confronto tra due operandi, tale confronto genera un valore numerico (1 confronto vero, 0 confronto falso)

· hanno tutti lo stesso livello gerarchico

· possono operare su variabili/costanti numeriche e/o carattere

· la variabile carattere viene trasformata in numerica se il confronto è tra numerica e carattere;

· il valore missing è sempre considerato il più piccolo

OPERATORI LOGICI

	Priorità
	
	

	1
	^
	NOT

	2
	&
	AND

	3
	|
	OR


Gli operatori logici consentono di mettere in relazione:

· più variabili

· due o più espressioni operando sul loro risultato 

LINGUAGGIO SAS: ISTRUZIONI “WHERE” e “IF..THEN..ELSE”

· Istruzione WHERE

· Istruzione IF..THEN <ELSE>

WHERE: seleziona le osservazioni in fase di esecuzione della procedura, lasciando inalterato l’archivio da cui legge i dati. Da ricordare che tale istruzione viene eseguita dopo che hanno avuto effetto le opzioni relative al Sas data set.

WHERE espressione;

IF:  valuta un’espressione e condizionatamente al risultato esegue i comandi che seguono

	IF expression THEN clause <; ELSE clause> 


/*uso di operatori aritmetici, comparazione, logici, WHERE, IF THEN ELSE*/
data info_capof1;set info_capof;

etasq1=eta1**2; *assegnazione;

*aggrgazione di modalità;

statociv1r=statociv1;

if statociv1=3 or statociv1=4 or statociv1=5 then statociv1r=3;

tit1r=titstu1;

if titstu1 in (1,2,3) then tit1r= 1;

if titstu1 =4 then tit1r= 2;

if titstu1 in(5,6) then tit1r= 3;

if titstu1 in(7,8) then tit1r= 4;

if posprof1 ge 1 and posprof1 le 9  then dip1=1; else dip1=0;

/*creazione di variabili dummy*/

Coniugato=0;

if statociv1=2 then coniugato=1;

High_edu=0;

if titstu1 le 2 then High_edu=1;

run;

/*uso di IF THEN per la creazione di nuovi data set*/
data  donne uomini;set info_capof;

if genere1=2 then output donne;else output uomini;

run;
Il FORMAT può essere definito anche per la visualizzazione di tabelle

********PROC FORMAT********;
PROC FORMAT ;

value genere 1='maschio'
             2='femmina';

value titstu 1='laurea o laurea breve'


       2='diploma'


       3='licenza media o qualifica'


       4='lic. elementare, analfabeta';

proc freq data=info_capof1;

tables genere1 tit1r;

format genere1 genere. tit1r titstu.;

run;

Conversione di variabili di un Sas dataset Numerica↔Carattere

Per convertire una variabile da numerica ad alfanumerica bisogna usare l'istuzione PUT come segue:

new-variable=PUT(old-variable,format)

data new;

set sashelp.class;

new=put(age,8.);

run;

Per convertire una variabile da alfanumerica a numerica  bisogna usare l'istuzione INPUT come segue:

new-variable=INPUT(old-variable,format)

data new2;

set new;

new2=input(new,best12.);

run;

Trucchetto:

A variabile carattere, X variabile numerica

Se si usa A con un operatore che richiede operandi numerici Y=A+X, SAS converte A in numerica;

Se si usa un operatore di comparazione, Y=A<X, la variabile carattere è convertita in numerica;

